
Séance 2 : Prise en main de SAS

Exécution d’un programme SAS

Un traitement est composé de 4 éléments (onglets du haut)
· PROGRAMME : le programme (ensemble d’instructions)

· JOURNAL : la description de l’exécution et la vérification des instructions

· RESULTATS : les résultats demandés (si le programme ne contient pas d’erreurs)

· DONNEES DE SORTIE : les tableaux créés (type feuille de calcul)
Quelques bonnes pratiques :


Toujours consulter JOURNAL avant de consulter RESULTATS


Vérifier que les données sont bien lues


Regarder s’il n’y a pas eu d’erreurs détectées

Un programme s’exécute  :


Par la fonction F8


En lançant « exécuter » (petit bonhomme)


On peut exécuter seulement une partie du code en la sélectionnant avec la souris puis F8

Un tableau de données SAS (Data)

Il se présente comme une feuille de calcul

· Chaque ligne représente « un individu »

· Chaque colonne représente « une variable »

Il peut faire l’objet de manipulation directe (sans programme) comme avec SPSS

Une instruction SAS

Une instruction se termine toujours par un point-virgule « ; ».

· Elle peut être sur plusieurs lignes

· Il peut y avoir plusieurs instructions sur une ligne

· SAS ne différencie pas les Minuscules et les Majuscules

Une instruction est analysée de manière interactive

· Couleur bleue : mot-clé reconnu par SAS

· Couleur rouge sombre : des libellés et étiquettes entre crochet (« ) et ( ») 

· Couleur verte : des commentaires

· Couleur noire : le reste

Title3 H=8 "Croisement genre et X";

      /* H hauteur de 8 points*/
Structure d’un programme SAS

Un programme est un ensemble 

· d’étapes DATA qui manipulent des tableaux de données

· de PROCédures qui traitent les données d’un tableau

· dont l’exécution est lancée par un « RUN ; »

Data : Variable

Une variable est, par défaut, définie uniquement à l’intérieur de son tableau (data). 

· dont l’exécution est lancée par un « RUN ; »

· Le nom de variable commence toujours par une lettre, contient des lettres ou des chiffres et le caractère souligné (_) mais pas d’espace.

· Ne pas commencer le nom des variables par un souligné "_" car les noms système commencent par ce caractère (_N_,…).

· Les variables sont, par défaut, des variables numériques. Elles peuvent aussi être des alphanumériques ou des dates.

Le groupage permet d’invoquer plusieurs variables de manière synthétique

· le tiret permet de traiter des variables ayant le même préfixe (Q1-Q5) 

· le double tiret permet de traiter les variables en séquence selon l’ordre de leur lecture (vendeur-- age) correspond à toutes les variables entre vendeur et age), 

· vendeur – numeric – age ne reprend que les variables numériques de la liste 
· (autre option – character- pour le texte)

· le deux points ":" remplace tous les suffixes (« X: » correspond à xage, xgenre, xmontant)

Data : Créer un data

Un tableau peut être créé de différentes manières.

· Dans le programme, 

· par une instruction INPUT

· qui lit des données suivant l’instruction CARDS ; et avant le point-virgule final

· Attention ! bien vérifier le nombre de variables car sinon SAS lit la valeur qui lui manque à la ligne suivante !

Data IN ;


/* Etape Data : nouveau tableau*/




/*« IN » est le nom donné au tableau*/ 

Input Nom $ Age ;

/*Lecture de 2 variables dont la première est alphabétique*/
Cards ;


/* Les données suivent */
Jules 12

Jacques 52

;
Run ;
Proc Print Data= in ;
/* Procédure : impression de vérification*/
Var Age ; 



Run ;
· A partir d’un autre tableau défini précédemment, 

· par l’instruction SET  (à partir de …)

Data Tableau_1 ;

/* Création d’un second tableau nommé Tableau_1 */

Set in ;

/* à partir des données du premier tableau  */
Age_2 = Age * Age ; 
/* Manipulation des données, ici mise au carré  */
Run ;
· A partir d’un fichier externe excel (xls) 
/* attention plusieurs lignes mais une seule instruction ( ;) */
Proc IMPORT datafile=‘C:/sasref/fichier_1.xls’/* adresse physique du fichier*/
out=getname_ex



/* nom du data  */
dbms=xls




/* Type de fichier : ici excel*/
replace; 




/* Remplacer si le fichier existe déjà */
getname=yes;

/* Utiliser les noms des colonnes pour les variables */



/* Attention ! tous les noms doivent exister */
run;

· A partir d’un fichier externe de données texte avec séparateur (délimiteur DLM) 

· Où les variables sont séparées par des délimiteurs

· Un « ; » (CSV), un « / », etc

Data in ;

infile 'c :\fichier_txt' 

/* adresse physique du fichier*/
dlm=' ' ;


/* indication du délimiteur*/
input var1 var2 $ ;

run ;

· A partir d’un fichier externe de données texte 

· Avec un dessin d’enregistrement (variable toujours à la même place, de longueur fixe
· Par défaut numérique, indiquer « $ » pour alphanumérique
· Position physique : 13-15 de la colonne 13 à la 15 (3 caractères)

· Par taille 3. = lire 3 caractères

· « +3 » sauter 3 colonnes

Data in ;

infile 'c :\fichier_txt' ;

/* adresse physique du fichier*/
input 
var1 3.

/* variable à la longueur connue*/
var2 $ 13-15 ;
/* variable alphanumérique à la position connue col13 à col 15*/
run ;

· A partir d’une procédure

· Avec une instruction du type OUTPUT ou OUT
· Utile par exemple avec l’analyse factorielle ou la typologie

Proc SORT  DATA = in OUT= in_trie; 
     
* crée un tableau trié (in_trie) ;

           By no_cli ; 


Exercice : dans le cadre d’une expérimentation (3x3) on a manipulé (OFFRE) la présence et le niveau (remboursement ou remboursement double) d’une garantie de satisfaction pour 3 niveaux de prix (PRIX : 1.62 1.92 2.22 €). Vous avez le numéro du groupe, le niveau de garantie, le prix, l’intention d’achat de la marque sans le prix (1 à 5 : forte), la cherté perçue (1 à 5 : très cher), l’intention d’achat avec annonce du prix et le niveau du prix habituellement payé pour un pack de 1 litre de jus d’orange frais.

· Lire http://www.mastermarketingdauphine.com/charge/ADD/Orange_mini.txt
· Vérifier que le tableau est bien créé;
Vérification des données

Vérification des données 

TOUJOURS bien s’assurer que les données ont été bien lues

Comment ?


En regardant les tableaux créés par la lecture (onglet « Données de sortie »)

Pourquoi ?

· Difficulté 1

· Un enregistrement = des données pour 1 individu et peut contenir plusieurs lignes

· Mais une « ligne » contient au maximum 256 caractères et se termine par un caractère de fin de ligne. Le programme cherche à lire les informations sur la ligne suivante !

· Bien vérifier que les données sont bien lues (en comparant avec les données brutes)

· Identification du problème : des données aberrantes et la réduction du nombre d’observations

· Difficulté 2 : séparateur des décimales  « , » en français mais « . » en US. Faire « Remplacer »

· Difficulté 3 : le mode de séparation des valeurs. Si on coupe/colle à partir d’excel, le séparateur est une tabulation et non un espace. Faire « Remplacer ».

Data/* : Commenter
Mettre des commentaires dans votre programme

Les commentaires ne sont pas des instructions. Ils sont ignorés lors de l’exécution.

· Pour vous souvenir de ce que vous voulez faire

· Pour que d’autres puissent reprendre votre travail

· Pour séparer les parties de programme ayant des fonctions différentes

Il y a deux types de commentaires

· Les commentaires « instructions » : toute instruction qui commence par « * » est ignoré :

·  * commentaire ;
· Les commentaires « neutralisant » : Tout ce qui se situe entre « /* » et « */ » est ignoré par le programme : /* commentaire */ 
· Ce dernier peut être utilisé « DANS » une instruction (avant le « ; » final)

* commentaire ;:  comme une ligne spécifique, débute par « * » et finit par « ; »
/* commentaire */ ; a l’intérieur d’une ligne : encadré par « /* » et « */ »



/* comme un bloc de plusieurs lignes que l’on veut rendre « inertes » pour le programme */

/*

Proc print data=in : run ;

*/
Proc print /* attention : changer le nom du data */ 

Data= in ;
Votre programme doit être « renseigné » et lisible.
/* =========================*/

/*Titre : premier programme */

/* date :                   */

/* objectif                 */

/* auteur                   */

/*version                   */

/* =========================*/

********************* ;

* Lecture des données ;

********************* ;

Data/Title : Renseigner les sorties

Mettre des titres et sous-titres

Permet de mettre des indications sur les documents de sortie

· Soit permanent (auteur, objet, date)

· Soit temporaire (donner un titre à l’analyse et préciser les données sur lesquelles les calculs sont effectués par exemple)

· Il existe plusieurs niveaux qui peuvent être utilisés conjointement

· Options : H pour la taille de la police et J pour la justification du texte

Deux types de titres sont utilisables

· En haut de page : TITLE

· En bas de page : FOOTNOTE

Title1 J=L "Mission Student"; 

Title2 H=2 J=L "Etude 1"; 




/* J= Justify=left ou right */
Title3 H=8 "Croisement genre et X";

      /* H hauteur de 8 points*/
Data/Label : Libeller les variables

Donne une valeur plus explicite au nom d'une variable pour faciliter l’interprétation.

· A mettre en fin d’une étape DATA avant le point-virgule
LABEL  Q1 = “Quantites”  
 

       Q2 = “Date commande”;

Data/Length : Fixer la taille des variables

Détermine la taille d’une variable en nombre de caractères

· Par défaut, la taille des variables alpha-numériques est limitée à 8 caractères ou à la taille de la première valeur lue. 

· Pour éviter de tronquer, la taille de la variable peut être définie dès le début de l’étape Data.

DATA in ;

       LENGTH Prenom $ 15 ;

Data/Attrib : Taille les variables 

Définit toutes les caractéristiques des variable (libellé, format, taille) 
· Astuce : une déclaration des variables dans ATTRIB permet d'en contrôler l'ordre pour les manipuler ensuite par groupe.

ATTRIB  Q1 LABEL= “Quantites”  

FORMAT= 4.1 length=6 

        Q2 LABEL= “Date commande P2 ”
FORMAT= DATETIME22.2 ;

Exercice : 

· renseigner votre programme

· créer les labels et les formats

=======================ETAPES PROC=======================
Proc SORT : Trier les informations

Trie les lignes d’un tableau 

· Le tri est une opération qui prend beaucoup de temps pour les gros fichiers.
· Le résultat du tri est stocké dans un nouveau tableau

· Le tri par défaut est croissant mais peut être spécifié décroissant (descending).

· Les formats sont « déclarés » une fois pour toute dans le programme et sont utilisables ensuite 

PROC SORT  DATA = in              
* précise le tableau à traiter ;

           OUT= intrie; 
     
* crée un tableau trié (intrie) ;

           By descending no_cli ; 
* tri des observations décroissant,

                                    (par défaut croissant) ;

A l’issue d’un tri (By nom_var), SAS définit deux variables (first.nom_var last.nom_var) qui définissent le premier et le dernier enregistrement de chaque modalité du tri. Ainsi un fichier de commandes trié par numéro de clients permettra d’accéder à la première et à la dernière commande de chaque client.

DATA tab_trie ; Set Non_trie ;

By no_cli ;

Retain cum_com cum_ca ;

If first.no_cli then do ; cum_co=1;         cum_ca=ca ;          end;

                Else do ; cum_co=cum_co+1 ; cum_ca=cum_ca + ca ; end;

BY itération d’une procédure selon une variable

Après le tri selon une variable, il est possible de faire exécuter plusieurs fois (résultats sur des pages différentes) une procédure selon les modalités de la variable triée.

PROC SORT data=in ;


By ville ;

PROC PRINT ;


By ville ; 


Var Popul ;

Proc PRINT : Imprimer les informations

Imprime pour chaque individu (ligne) les variables indiquées après VAR ;

· Les formats sont « déclarés » une fois pour toute dans le programme et sont utilisables ensuite dans les procédures

Proc PRINT ; 

        Var  X1 X3; 
      * définit les variables et l’ordre

Possibilité de « répéter » l’instruction selon les modalités de la variable suivant  «BY ».

· Le data doit d’abord être trié selon cette variable

Proc SORT data=in ; by dept ; run ;

Proc PRINT ; 

        By 
departement ;  
* section séparée selon le dépt ;






* trier au préalable sur la variable ;

        Pageby vendeur

* saut de page forcé à chaque vendeur ;

        ID
nom


* remplace le numéro par le nom ;


        Sum  X2


* ajoute le total ;

        Var  X1 X2 X3; 
       * définit les variables et l’ordre

Options utiles pour la procédure PRINT :

· Définir la première observation (FIRSTOBS=) 

· Le nombre d’observations à imprimer (OBS=)

· L’impression sur un critère de sélection (WHERE=(sexe= « H »))

Proc print  data=in (fisrtobs=10 obs=20 );
Proc print  data=in (where=(sexe=’H’ )) ;     * impression selective ;

Proc FORMAT : Formater les modalités
Donner une étiquette à chaque modalité d’une variable catégorielle
· Les formats sont « déclarés » une fois pour toute dans le programme et sont utilisables ensuite dans les procédures

· Un format est une étiquette qui apparaît dans les sorties au lieu du code (un masque)
· Une ligne « value » donne un nom et définit le format d’une variable 

· Le nom du format est libre : le nom de la variable elle-même ou un nom qui fasse sens (par exemple variable suivie de « _fmt » pour format)

· Pour une variable catégorielle, il permet de ne stocker que le code (H au lieu de Homme) tout en ayant des sorties facilement lisibles

· Pour une variable continue, il permet  de catégoriser facilement et de manière temporaire une variable continue : moins de 14  = « Petit » et au delà = « Grand »

· Un format de variable alphanumérique commence par un dollar ($)

Proc Format ;

value $genre_fmt "H"= "Homme" "F"= "Femme" ; 
/* $ pour un alphanumérique */
value x1_fmt     1-14= "Petit" 15-high= "Grand"  ; 
/* low, high, other, aussi 1,3,24  */
value x2_fmt     1- 5= "Small" 6-high = "Large"  ; 

run; 

· Le format est utilisé dans les procédures, sur une ligne supplémentaire regroupant les formats des variables utilisées dans la procédure

· Chaque variable est suivie du nom du format que l’on veut utiliser SUIVI d’UN POINT (« . »).

Proc Freq data=in ;

Table taille*genre ;

Format Genre $genre_fmt. taille x1_fmt. ;

run; 

Exercice : 

· Trier le fichier par Prix puis Offre
· Créer les formats

· Trier en prix de référence décroissant
· Imprimer les 10 premières observations
Proc FREQ : Tableau de fréquences

Comptage des effectifs (« tris »)
Les fréquences simples sont le comptage du nombre d’observations ayant chaque modalité.

· Elles sont absolues (effectifs) ou relatives (en pourcentage : ligne, colonne, total), par modalité ou cumulées

· Il est possible de trier selon la fréquence décroissante en spécifiant en option ORDER=FREQ.

proc Freq data=in ;

table

genre 


/* tri simple - fréquences simples */
run;

Un tableau de contingence est un tableau croisé des fréquences de deux (ou plusieurs) variables

· En standard, les pourcentages lignes, colonnes et case apparaissent. Ils peuvent être supprimés avec NOROW, NOCOL et NOPERCENT.

proc Freq data=in ;

table

genre*x ;

/* tableau de contingence (tri / tableau croisé) */
run;

proc Freq data=in  ; 

/* en sélectionant les pourcentages souhaités */ 

table genre*x  

/ nocol nopercent;   
/* aussi norow */ 

run;
proc Freq data=in  ; 

/* avec un format */
 

table genre*x  

/ nocol nopercent;   

/* aussi norow */ 

format x x_fmt. genre $genre_fmt. ; /* ATTENTION au point «.» après fmt */
run;
Plusieurs demandes de croisement peuvent se succéder avec des regroupements :

proc Freq data=in ;

table

X1*X2 X1*X3 /* */
table

(X1 X2)*(X3 X4) ;

/* X1*X3 X1*X4 X2*X3 X2*X4 */
table

X1*X2*X3; 


/* toutes les combinaisons 2 à 2*/
run;

· Pour une variable continue, l’instruction FORMAT X1 5. crée des modalités avec un pas de 5, sans cette instruction toutes les valeurs seront des modalités !
Exercice : à partir de la base “orange.mini.txt”, 

· vérifier les effectifs par groupe

· faire un croisement intention d’achat sans prix et avec prix avec uniquement les effectifs (sans %)
· faire un croisement cherté perçue selon le niveau de prix manipulé

· idem avec seulement les % lignes
Proc TABULATE :Tableau de synthèse

Calculs effectués pour chaque classe d’individus

· Procédure qui permet de faire un tableau de valeur de synthèse pour des individus classés selon plusieurs dimensions. 

· CLASS déclare les variables catégorielles utilisées pour la ventilation des résultats mais sur lesquelles aucun traitement arithmétique ne sera effectué ;

· VAR indique les variables sur lesquelles des calculs seront faits ;

· By répète éventuellement la procédure séparément pour une autre variable (tableaux séparés « par » région, pays, …;

· TABLE décrit la manière dont le tableau est constitué : les deux dimensions (lignes, colonnes) sont séparées par une virgule (pour plus de clarté ici sur deux lignes). 
· Les calculs : N (effectifs), Mean (moyenne), Max, Min, Std (écart-type),… 

· Ils sont effectués pour la variable précédent l’« * »

PROC Tabulate data=in;

Class dept ville ;


/* variables de classification */

Var Popul resid revenu ;

/* variables de calcul */

table (ville all), 

/* variables en lignes (avant la virgule« , »*/
dept*(N (popul resid)*sum /* variables en colonne */
; 

· Les deux dimensions, ligne ou colonne, peuvent comporter des variables catégorielles ou des variables de calcul
· Les variables de classification peuvent être imbriquées (X1*X2 : toutes les modalités de X2 pour chaque modalité de X1)

· l’ajout de ALL permet d’avoir des totaux par modalité (X1*(X2 All) : ajout du total pour chaque modalité de X1)

· Pour chaque variable de calcul, le format précise éventuellement le nombre total de caractères et le nombre de chiffres derrière la virgule (X1*F=10.3)

· La longueur du libellé des lignes peut être fixée par « / RTS = 10 »
PROC Tabulate data=in;

Class dept ville ;

Var Popul resid revenu ;

table (ville all)*(N (popul resid)*sum*F=10.0 /* Format  */
revenu*(mean std max min)*F=6.1 ),
(dept all)

/RTS=10; 

/* libellé de la ligne en 10 caractères */
Exercice : à partir de la base “orange.mini.txt”, 

· Calculer la moyenne et l’écart-type de la cherté perçue pour chaque groupe expérimental (avec 2 chiffres derrière la virgule)
· Faire un tableau avec les mean, min et max du prix de référence

· Faire les éventuelles corrections
Proc RANK : Créer des rangs

Passer d’une variable au minimum ordinale, à une variable ordinale en « n » classes.

· CLASS déclare les variables catégorielles utilisées pour la ventilation des résultats mais sur lesquelles aucun traitement arithmétique ne sera effectué ;

· Le rang est créé dans un nouveau tableau qui reprend les autres variables. Si aucun tableau n’a pas été spécifié dans out=  , les résultats sont disponibles dans un nouveau tableau DATAn.

· L’option de groupage (GROUP=) permet de constituer des groupes d’effectifs à peu près identiques (quartiles, déciles, percentiles,…). Ici 5 groupes sont constitués pour les 3 variables (montant, fréquence et dernier)

· Les rangs sont stockés dans les variables R_montant R_freq R_dernier (RANKS). Sans la spécification de ranks, la variable garde le même nom et les rangs viennent écraser l’information. .  

Proc RANK data=in group=5 out= in_R;

Class dept ville ;

Var 
montant freq dernier;

/* variables à traiter */
Ranks 
R_montant R_freq R_dernier;
/* variables de rang créées */
Run ;
Exercice : à partir de la base “orange.mini.txt”, 

· Créer une variable R_PRef en classant les prix de référence en quartiles

· Calculer la moyenne de la cherté perçue pour chaque modalité de R_PRef (en ligne) et chaque niveau de prix (en colonne)
Solution des exercices

http://www.mastermarketingdauphine.com/charge/ADD/Orange_mini.sas
*************;
* Exercice 1 ;
*************;
/* orange mini */
/* Plan expérimental offre 3 x prix 3= */
/***************************************/
data in ; 



input Groupe Offre Prix IA0 cherte IA_prix Prix_reference ;

num= _N_ ; /* création d'un numéro de questionnaire */
Label 


Prix 
= "Prix"

IA0

= "Intention d’achat sans prix "

Cherte
= "Cherté perçue"

IA_prix
= "Intention d’achat avec prix "

Prix_reference= "Prix habituel paye pour un jus d’orange";

cards ;

1 0 1 5 3 2 2.1 

1 0 1 3 2 5 2.10 

1 0 1 7 4 7 1.59 

1 0 1 6 2 2 1.80 

1 0 1 5 3 2 1.9 

1 0 1 6 3 5 2 

…

9 2 3 3 3 1 2.35 

9 2 3 7 5 7 1.8 

;

run;

title1 "Expérimentation Jus d’orange";

title3 "Vérification des effectifs";

Footnote1 J=l H=1 "Master Marketing";

*************;
* Exercice 2 ;
*************;
Proc format ;


value prix_fmt 
1="1.62" 2="1.92" 3="2.22" ;


value offre_fmt 0="sans garantie" 1="Garantie simple" 2="Garantie double";

run ;

title3 "Impression des 10 premières observations";

Proc print data=in (Obs=10);

Run ;

title3 "Tri décroissant sur le prix de référence";

Proc sort data=in out=in_trie;

by descending prix_reference ;

Run ;

title3 "Impression des 10 premières observations du fichier trié";

Proc print data=in_trie (Obs=10);

Run ;

title3 "Correction de la valeur erronée";

Data in; set in ;


If Prix_reference= 195 then prix_reference=1.95;

run ;

Proc print data=in (Obs=10);

Run ;

title3 "Vérification des effectifs";

proc freq data=in;

table groupe ;

table offre * prix 


/ nocol norow nopercent;

format offre offre_fmt. prix prix_fmt. ;

run ;

title3 "Croisement avant après";

proc freq data=in;

table IA0 * IA_prix 


/ Nocol Norow Nopercent;

run ;

title3 "Croisement Prix Cherté perçue";

proc freq data=in;table groupe ;

table prix * cherte


/nocol nopercent;

format Prix prix_fmt.;

run ;

*************;
* Exercice 3 ;
*************;
title3 "Influence du prix sur la cherté perçue";

proc tabulate data=in;

class prix ;

var cherte ;





table prix, cherte*(mean std max min)*F=4.2 ; /* lignes , colonnes */
format Prix prix_fmt.;

run ;

title3 "Rangs du prix de reférence";

proc Rank data=in group=4 Out=R_In;

var prix_reference ;

ranks R_PRef ;

Run ;

proc tabulate data=R_in;

class R_PRef ;

var prix_reference ;





table R_PRef, prix_reference*( N*F=5.0 mean*F=4.2) ;

run ;

title3 "Influence du prix sur la cherté perçue";

proc tabulate data=R_in;

class R_PRef Prix;

var Cherte ;





table R_PRef, Prix*Cherte*( mean*F=4.2) ;

format Prix prix_fmt.;

run ;

